unittest
{
	assert( [ "2 60", "10 10", "100 100" ].parse.expand.solve == 110 );
	assert( [ "3 60", "10 10", "10 20", "10 30" ].parse.expand.solve == 60 );
	assert( [ "3 60", "30 10", "30 20", "30 30" ].parse.expand.solve == 50 );
	assert( [ "10 100", "15 23", "20 18", "13 17", "24 12", "18 29", "19 27", "23 21", "18 20", "27 15", "22 25" ].parse.expand.solve == 145 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nt = input.front.split.to!( long[] );
	auto abs = new long[ 2 ][ nt[ 0 ] ];
	foreach( ref ab; abs )
	{
		input.popFront;
		ab = input.front.split.to!( long[] );
	}
	return tuple( nt[ 1 ], abs );
}

auto solve( long t, long[ 2 ][] abs )
{
	// 時間内に全部頼んで食べ終われるならそれで良し
	auto mt = 0L;
	auto md = 0L;
	foreach( ab; abs )
	{
		mt += ab[ 0 ];
		md += ab[ 1 ];
	}
	if( mt <= t ) return md;
	
	// dp[ 経過時間 ] = 美味しさの合計
	auto dp = new long[ 1 + t ];
	dp[ 1 .. $ ] = -1;
	foreach( ab; abs.sort() )		// 食べる時間が一番長いやつを最後に持ってくるべきなので必要時間が小さい順に試す
	{
		foreach_reverse( long tt; 0L .. t )
		{
			if( dp[ tt ] < 0 ) continue;
			
			auto k = min( t, tt + ab[ 0 ] );		// ラストオーダーに間に合えば食べる時間がオーバーしても良い
			dp[ k ] = max( dp[ k ], dp[ tt ] + ab[ 1 ] );
		}
	}
	return dp[ $ - 1 ];
}
